# $Id: PKGBUILD 160233 2012-05-31 11:55:41Z bisson $
# Contributor: damir <damir@archlinux.org>
# Contributor: Thayer Williams <thayer@archlinux.org>
# Contributor: Daniel J Griffiths <ghost1227@archlinux.us>
# Maintainer: Gaetan Bisson <bisson@archlinux.org>

pkgname=abook
pkgver=0.6.0pre2
pkgrel=5
pkgdesc='Text-based addressbook designed for use with Mutt'
arch=('i686' 'x86_64')
url='http://abook.sourceforge.net/'
license=('GPL2')
depends=('readline')
source=("http://abook.sourceforge.net/devel/abook-${pkgver}.tar.gz"
        'vcard-compat.patch')
sha1sums=('42a939fba43e51aa011fa185113c12ec4bc1e1ec'
          'feb35deb79116c187953d06e514c7918be5646b6')

build() {
	cd "${srcdir}/${pkgname}-${pkgver}"
	patch -p1 -i ../vcard-compat.patch
	./configure --prefix=/usr --mandir=/usr/share/man
	make
}

package() {
	cd "${srcdir}/${pkgname}-${pkgver}"
	make DESTDIR="${pkgdir}" install
}
